Friedmannove jednadžbe

Ovo je prilagodba radnog listića dostupnog ovdje (http://sagemanifolds.obspm.fr/examples.html).

Da bi sve funkcioniralo, osim jupyter notebooka potrebno je instalirati i Sage kernel. To se na Arch linuxu može tim redom. Drugdje je možda pogodnije naprosto instalirati Sage koji mislim dolazi sa svojim jupyter notebookom.

In [1]:
```
version()
```

Out[1]: 'SageMath version 8.1, Release Date: 2017-12-07'

Postavljanje prikaza simbola korištenjem LaTeX formatiranja:

In [2]:
```
%display latex
```

Deklariramo prostorvrijeme kao 4-dimenzionalnu diferencijabilnu mnogostrukost M:

In [3]:
```
M = Manifold(4, 'M')
print(M)
```

```
4-dimensional differentiable manifold M
```

Uvodimo standardne (FL)RW koordinate, koristeći metodu `chart()`, s argumentom koji je python string s razmaknuto navedenim koordinatama u sintaksi `simbol:raspon vrijednosti:LaTeX simbol` gdje je defaultni raspon vrijednosti $(-\infty, \infty)$:

In [4]:
```
fr.<t,r,th,ph> = M.chart(r't r:[0,+oo) th:[0,pi]:\theta ph:[0,2*p
i):\phi')
fr
```

Out[4]: $(M, (t, r, \theta, \phi))$

Definiramo skalarne varijable: Newtonovu konstantu G, konstantu prostorne zakrivljenosti $k \in \{1, 0, -1\}$, radijus zakrivljenosti R_0, faktor skale $a(t)$, gustoća kozmičkog fluida $\rho(t)$ i njegov tlak $p(t)$. Za RW metriku koristimo izraz iz Ryden (2016)

$$ds^2 = c^2 dt^2 - a^2(t) \left(\frac{dr^2}{1 - \frac{kr^2}{R_0^2}} + r^2 d\Omega \right)$$

uz napomenu da koristimo suprotni preznak metrike i da većina literature apsorbira radijus zakrivljenosti R_0 ili u konstantu zakrivljenosti $K \equiv k/R_0^2$ ili u faktor skale uz redefiniciju koordinate $r \to rR_0$ tako da postane bezdimenzionalna.

In [5]:
```
var('G, Lambda, k, R0, c', domain='real')
a = M.scalar_field(function('a')(t), name='a')
rho = M.scalar_field(function('rho')(t), name='rho')
p = M.scalar_field(function('p')(t), name='p')
```

RW metriku definiramo u "-" konvenciji (tj. "čestičarska" ili "West coast", dakle obrnuto od Ryden):

In [6]:
```
g = M.lorentzian_metric('g', signature='negative')
g[0,0] = c*c
g[1,1] = -a*a/(1 - k*r^2/R0^2)
g[2,2] = -a*a*r^2
g[3,3] = -a*a*(r*sin(th))^2
g.display()
```

Out[6]:

$$g = c^2 \mathrm{d}t \otimes \mathrm{d}t + \left(\frac{a(t)^2}{\frac{kr^2}{R_0^2} - 1} \right) \mathrm{d}r \otimes \mathrm{d}r - r^2 a(t)^2 \mathrm{d}\theta \otimes \mathrm{d}\theta - r^2 a(t)^2 \sin(\theta)^2 \mathrm{d}\phi$$

A matrix view of the metric components:

In [7]:
```
g[:]
```

Out[7]:

$$\begin{pmatrix} c^2 & 0 & 0 & 0 \\ 0 & \frac{a(t)^2}{\frac{kr^2}{R_0^2} - 1} & 0 & 0 \\ 0 & 0 & -r^2 a(t)^2 & 0 \\ 0 & 0 & 0 & -r^2 a(t)^2 \sin(\theta)^2 \end{pmatrix}$$

Christoffelovi simboli su kvadratni u metričkom tenzoru pa ne ovise o konvencijama za predznake:

In [8]:
```
nabla = g.connection()
g.christoffel_symbols_display()
```

Out[8]:

$$\Gamma^{t}{}_{r\,r} = -\frac{R_0^2 a(t) \frac{\partial a}{\partial t}}{c^2 k r^2 - R_0^2 c^2}$$

$$\Gamma^{t}{}_{\theta\,\theta} = \frac{r^2 a(t) \frac{\partial a}{\partial t}}{c^2}$$

$$\Gamma^{t}{}_{\phi\,\phi} = \frac{r^2 a(t) \sin(\theta)^2 \frac{\partial a}{\partial t}}{c^2}$$

$$\Gamma^{r}{}_{t\,r} = \frac{\frac{\partial a}{\partial t}}{a(t)}$$

$$\Gamma^{r}{}_{r\,r} = -\frac{kr}{kr^2 - R_0^2}$$

$$\Gamma^{r}{}_{\theta\,\theta} = \frac{kr^3 - R_0^2 r}{R_0^2}$$

$$\Gamma^{r}{}_{\phi\,\phi} = \frac{\left(kr^3 - R_0^2 r\right) \sin(\theta)^2}{R_0^2}$$

$$\Gamma^{\theta}{}_{t\,\theta} = \frac{\frac{\partial a}{\partial t}}{a(t)}$$

$$\Gamma^{\theta}{}_{r\,\theta} = \frac{1}{r}$$

$$\Gamma^{\theta}{}_{\phi\,\phi} = -\cos(\theta)\sin(\theta)$$

$$\Gamma^{\phi}{}_{t\,\phi} = \frac{\frac{\partial a}{\partial t}}{a(t)}$$

$$\Gamma^{\phi}{}_{r\,\phi} = \frac{1}{r}$$

$$\Gamma^{\phi}{}_{\theta\,\phi} = \frac{\cos(\theta)}{\sin(\theta)}$$

Riccijev tenzor:

In [9]:
```
Ricci = nabla.ricci()
Ricci.display_comp()
```

Out[9]:

$$\mathrm{Ric}(g)_{t\,t} = -\frac{3 \frac{\partial^2 a}{\partial t^2}}{a(t)}$$

$$\mathrm{Ric}(g)_{r\,r} = -\frac{2 R_0^2 \left(\frac{\partial a}{\partial t}\right)^2 + R_0^2 a(t) \frac{\partial^2 a}{\partial t^2} + 2 c^2 k}{c^2 k r^2 - R_0^2 c^2}$$

$$\mathrm{Ric}(g)_{\theta\,\theta} = \frac{2 R_0^2 r^2 \left(\frac{\partial a}{\partial t}\right)^2 + R_0^2 r^2 a(t) \frac{\partial^2 a}{\partial t^2} + 2 c^2 k r^2}{R_0^2 c^2}$$

$$\mathrm{Ric}(g)_{\phi\,\phi} = \frac{\left(2 R_0^2 r^2 \left(\frac{\partial a}{\partial t}\right)^2 + R_0^2 r^2 a(t) \frac{\partial^2 a}{\partial t^2} + 2 c^2 k r^2\right) \sin(\theta)^2}{R_0^2 c^2}$$

Ovi izrazrazi su, nakon zamjene $k \rightarrow kR_0$, u slaganju s wikipedijom (https://en.wikipedia.org/wiki/Friedmann%E2%80%93Lema%C3%AEtre%E2%80%93Robertson%E2%80%93Wall

$$R_{tt} = -3\frac{\ddot{a}}{a},$$

$$R_{rr} = \frac{c^{-2}(a(t)\ddot{a}(t) + 2\dot{a}^2(t)) + 2k}{1 - kr^2}$$

$$R_{\theta\theta} = r^2(c^{-2}(a(t)\ddot{a}(t) + 2\dot{a}^2(t)) + 2k)$$

$$R_{\phi\phi} = r^2(c^{-2}(a(t)\ddot{a}(t) + 2\dot{a}^2(t)) + 2k)\sin^2(\theta)$$

Iz dokumentacije SageManifolds paketa može se zaključiti da isti koristi "+" konvenciju za prezdnak Riemannovog tenzora, kao i "+" konvenciju za vezu Riccijevog i Riemannovog tenzora (gdje za razliku od gornje konvencije za metriku korisnik to ne može mijenjati), što znači da je nužno pisati Einsteinovu jednadžbu gravitacije u "+" konvenciji, tj.

$$R_{\mu\nu} - \frac{1}{2}g_{\mu\nu}R = +\frac{8\pi G}{c^4}T_{\mu\nu}$$

Riccijev skalar (${R^\mu}_\mu$):

In [10]:
```
Ricci_scalar = g.ricci_scalar()
Ricci_scalar.display()
```

Out[10]:

$$\begin{array}{llcl} \mathrm{r}\left(g\right): & M & \longrightarrow & \mathbb{R} \\ & (t, r, \theta, \phi) & \longmapsto & -\frac{6 \left(R_0^2 \left(\frac{\partial\, a}{\partial t}\right)^2 + R_0^2 a(t) \frac{\partial^2\, a}{\partial t^2} + c^2 k\right)}{R_0^2 c^2 a(t)^2} \end{array}$$

(Za razliku od Riccijevog tenzora koji je invarijantan na odabir konvencije za metrički tenzor, Riccijev skalar bi u drugoj konvenciji za metrički tenzor imao obrnuti predznak, npr. na gornjoj stranici wikipedije.) Uočavamo da je za statičko prostorvrijeme Riccijev skalar

In [11]:
```
Ricci_scalar.expr().subs({a.expr():1, diff(a.expr(),t):0, diff(a.expr(),t,2):0})
```

Out[11]:

$$-\frac{6\,k}{R_0^2}$$

dakle proporcionalan je Gaussovoj zakrivljenosti plohe (relativni minus je u ovoj konvenciji za metriku jer je zakrivljen prostorni dio). To je manifestacija Gaussovog "teorema egregium" po kojem je ekstrinsično svojstvo plohe (radijus zakrivljenosti) potpuno određeno intrinsičnom metrikom na plohi.

4-brzina kozmičkog fluida:

In [12]:
```
u = M.vector_field('u')
u[0] = 1/c
u.display()
```

Out[12]:
$$u = \frac{1}{c}\frac{\partial}{\partial t}$$

Tenzor energije-impulsa T za savršeni fluid:

In [13]:
```
u_form = u.down(g) # the 1-form associated to u by metric duality
T = (rho*c^2+p)*(u_form*u_form) - p*g
T.set_name('T')
print(T)
T.display()
```

```
Field of symmetric bilinear forms T on the 4-dimensional differentiab
le manifold M
```

Out[13]:
$$T = c^4\rho(t)\,\mathrm{d}t\otimes\mathrm{d}t + \left(-\frac{R_0^2 a(t)^2 p(t)}{kr^2 - R_0^2}\right)\mathrm{d}r\otimes\mathrm{d}r + r^2 a(t)^2 p(t)\,\mathrm{d}\theta\otimes\mathrm{d}\theta + r^2 a$$
$$\otimes\,\mathrm{d}\phi$$

Trag od T:

In [14]:
```
Ttrace = g.inverse()['^ab']*T['_ab']
Ttrace.display()
```

Out[14]:
$$\begin{array}{llcl} & M & \longrightarrow & \mathbb{R} \\ & (t,r,\theta,\phi) & \longmapsto & c^2\rho(t) - 3\,p(t)\end{array}$$

Einsteinova jednadžba gravitacije. Kako je gore navedeno, nužno u "+" konvenciji: $R_{\mu\nu} - \frac{1}{2}Rg_{\mu\nu} = 8\pi GT_{\mu\nu}$

In [15]:
```
E1 = Ricci - Ricci_scalar/2*g - (8*pi*G/c^4)*T
print("Friedmannova jednadžba:\n")
(E1[0,0]/3).expr().expand() == 0      # dividing everything by 3
```

```
Friedmannova jednadžba:
```

Out[15]:
$$-\frac{8}{3}\pi G\rho(t) + \frac{\frac{\partial}{\partial t}a(t)^2}{a(t)^2} + \frac{c^2 k}{R_0^2 a(t)^2} = 0$$

Trace-reversed version of the Einstein equation: $R_{\mu\nu} - \Lambda g_{\mu\nu} = 8\pi G \left(T_{\mu\nu} - \frac{1}{2} T\, g_{\mu\nu}\right)$

In [16]:
```
E2 = Ricci - (8*pi*G/c^4)*(T - Ttrace/2*g)
print("Jednadžba ubrzanja:\n")
(E2[0,0]/3).expr().expand() == 0
```

```
Jednadžba ubrzanja:
```

Out[16]:

$$-\frac{4}{3}\,\pi G\rho\left(t\right) - \frac{4\,\pi G p\left(t\right)}{c^2} - \frac{\frac{\partial^2}{(\partial t)^2}a\left(t\right)}{a\left(t\right)} = 0$$